# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=wingpanel-indicator-power
pkgver=6.2.1
pkgrel=1
pkgdesc='Wingpanel Power Indicator'
arch=(loong64 x86_64)
url=https://github.com/elementary/wingpanel-indicator-power
license=(GPL3)
groups=(pantheon)
depends=(
  bamf
  glib2
  gtk3
  libgee
  libgranite.so
  libgtop
  libudev.so
  libwingpanel.so
)
makedepends=(
  git
  intltool
  meson
  vala
)
_tag=59a227ff73967d55f2f6117f0568c0851dbd7636
source=(git+https://github.com/elementary/wingpanel-indicator-power.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd wingpanel-indicator-power

  git describe --tags
}

build() {
  arch-meson wingpanel-indicator-power build
  ninja -C build
}

package() {
  DESTDIR="${pkgdir}" ninja -C build install
}

# vim: ts=2 sw=2 et:
